Gear Design Options

William L. Janninck
ITW-Illitron Automated Systems
Lincolnwood, IL

AUTHOR:

W. L. JANNINCK *is a gear geometrician and consultant to the Illinois Tools Division of ITW Inc. He has experience in the design and application of both generating and forming type metal cutting tools, specializing in the area of tooling and gaging for the manufacture of gears, splines and sprockets. Prior to entering private practice, he was manager of product engineering and chief engineer at ITD. He has served on various commitees of the AGMA and MCTI and is chairman of the AGMA Cutting Tool Committee. He also served on the SAE-ANSI Committee on Power Transmission Chains and Sprockets. He has written extensively on tool applications, gaging, gear design and gear inspection. He was educated at Northwestern University, Evanston, Illinois.*

An Introduction to a Spur Gearset Design

When specifying a complete gear design, the novice designer is confronted with an overwhelming and frequently confusing group of options which must be specified. This array of specifications range from the rather vague to the very specific.

There are many ways to narrow the selection and guide the designer in selecting the less easily defined parameters. Some values, especially the geometric ones, must be specified very exactly. Others can be selected rather loosely on the basis of choice or experience. (See Table 1.)

One effective method of controlling and organizing this set of specifications is based on laying out a group of possible choices for consideration that first meet the geometrical requirements of center distance and gear ratio, and then considering the physical requirements of transmitted power, input speed, gear material and gear accuracy. The availability of the personal or micro-computer has also made it quite simple to compute a group of different designs that would be possible candidates for a final design.

Table 2—Design Choices for Single Pair Set of External Spur Gears

Center distance	5.0 inches
Gear ratio	4 to 1 reduction
Pinion speed	1800 RPM
Face width	.80 inches
Gear material	Steel
Hardness	300 BHN

To simplify the example and also to eliminate a few of the choices we will give an example of a design for a single pair set of external spur gears with the specifications shown below.

The face width was arbitrarily chosen, but was based on a scale layout of the anticipated set using the operating pitch circles. A common sense of proportion prevailed. The gear material picked was a typical machineable gear steel that can be finish cut, preserving the cut gear accuracy. Expected accuracy would be in the range of AGMA 7 or 8.

Using the gear ratio and the center distance as input, a short computer program is used to arrive at a number of choices for gear and pinion teeth and related diametral pitch. This program, (Program 1) starts with the pinion teeth ranging from 10 to 55, sequentially, computing the mating gear teeth closest to the desired ratio, and prints out the associated diametral pitch for standard operating conditions. This gives 46 different possible sets to choose from. A sample output is shown in Table 3.

The number of choices may be further reduced by selecting some sets where currently available cutting tools can be used or a scattered group can be arbitrarily selected, ranging

Table 1—Basic Gear Design Specifications

Type of gear spur or helical	Helix angle & hand	Type of teeth internal or external
Diametral pitch	Pressure angle	Pinion & gear teeth
Gear ratio	Tooth form	Pinion enlargement
Fillet form	Center distance	Tooth proportion
Face width	Material & treatment	Input speed
Strength	Quality level	Life expectancy
Noise	Manufacturing methods	Lubrication

Table 3 – Available choices of teeth for spur gear sets.

Center Distance = 5
Target Gear Ratio = 4

Pinion T.	Gear T.	Act. Ratio	Diam. Pitch
10	40	4.0000	5.00000
11	44	4.0000	5.50000
12	48	4.0000	6.00000
13	52	4.0000	6.50000
14	56	4.0000	7.00000
15	60	4.0000	7.50000
16	64	4.0000	8.00000
17	68	4.0000	8.50000
18	72	4.0000	9.00000
19	76	4.0000	9.50000
20	80	4.0000	10.00000
21	84	4.0000	10.50000
22	88	4.0000	11.00000
23	92	4.0000	11.50000
24	96	4.0000	12.00000
25	100	4.0000	12.50000
26	104	4.0000	13.00000
27	108	4.0000	13.50000
28	112	4.0000	14.00000
29	116	4.0000	14.50000
30	120	4.0000	15.00000
31	124	4.0000	15.50000
32	128	4.0000	16.00000
33	132	4.0000	16.50000
34	136	4.0000	17.00000
35	140	4.0000	17.50000
36	144	4.0000	18.00000
37	148	4.0000	18.50000
38	152	4.0000	19.00000
39	156	4.0000	19.50000
40	160	4.0000	20.00000
41	164	4.0000	20.50000
42	168	4.0000	21.00000
43	172	4.0000	21.50000
44	176	4.0000	22.00000
45	180	4.0000	22.50000
46	184	4.0000	23.00000
47	188	4.0000	23.50000
48	192	4.0000	24.00000
49	196	4.0000	24.50000
50	200	4.0000	25.00000
51	204	4.0000	25.50000
52	208	4.0000	26.00000
53	212	4.0000	26.50000
54	216	4.0000	27.00000
55	220	4.0000	27.50000

Program 1 – Tooth Selector Program

```
10 REM – Toothsel.bas
20 REM – This program develops a series of choices of gear
   teeth sets
30 REM – for a specific center distance and ratio desired.
40 KEY OFF:CLS
50 DIM NP (55) ,NG(55) ,ACTR(55) ,DP(55)
60 INPUT "Center Distance  ";CD
70 INPUT "Ratio Desired  "    ;RATIO
80 FOR I=10 TO 55
90 NP(I)=I:NG(I)=INT(I*RATIO+.5)
100 ACTR (I)=
   NG(I)/NP(I):DP(I)=(NP(I)+NG(I))/2/CD
110 NEXT I
120 LPRINT"     Available choices of teeth for spur gear
   sets."
130 LPRINT
140 LPRINT"     Center Distance = ",CD
150 LPRINT"     Target Gear Ratio = ",RATIO
160 LPRINT
170 LPRINT" Pinion T.  Gear T.     Act.
Ratio     Diam. Pitch"
180 FOR I=10 TO 55
190 Z$="     ##      ###      ###.####      ##.#####"
200 LPRINT USING Z$;NP(I) ;NG(I);ACTR(I);DP(I)
210 NEXT I
```

through various teeth numbers and pitches. For example, the sample list in this case contains commonly available pitches such as 5, 6, 8, 10, 12, 14, 16 and 20. If an odd target gear ratio or an uneven center distance is used, the possibility of standard diametral pitches is reduced.

If the teeth versus pressure angle chart for natural undercut shown in Fig. 1 is examined, the choice will be guided away from the lower pinion teeth numbers and lower pressure angles because natural undercutting causes poor gear operating conditions and also reduces strength.

In the example case the first likely choice is a 16-64 tooth gearset of 8DP and 20 or 25° pressure angle. The next choice would be 20-80 teeth of 10DP in 20 or 25° pressure angle, and so on thru 20DP.

Actually there is a great deal of flexibility in gear design and long addendum pinions and short addendum gears can be used to avoid or reduce undercutting. It is also possible to depart from the specified diametral pitch by a small amount, resulting in an oversize or undersize operating condition on center distance.

A flexible computer program was written to accomodate both standard and non-standard conditions and to freeze the total gear geometry for a particular gear set. The input values required to run the program are listed in Table 4.

The program has a standard basic rack embedded within it for 14.5 thru 25°PA as shown in Fig. 2. This basic rack is used, along with the circular tooth thickness on the pinion and the backlash desired in the set to do a complete gear design. If the pinion tooth thickness is set at one half of the



MINIMUM NUMBER OF GEAR TEETH WITHOUT UNDERCUT

Fig. 1

Table 4

40 INPUT "Diametral Pitch	= ",DP
50 INPUT "Pressure Angle	= ",PA
60 INPUT "Center Distance	= ",CD
70 INPUT "Face Width	= ",F
80 INPUT "Steel Allow Tens.	= ",SAT
90 INPUT "Steel Allow Comp.	= ",SAC
100 INPUT "Pinion RPM	= ",RPM
110 INPUT "Pinion Teeth	= ",NP
120 INPUT "Gear Teeth	= ",NG
130 INPUT "Pinion CTT	= ",CTP
140 INPUT "Backlash	= ",BL



PA	TIP FLAT
14.5°	.924/DP
20°	.661/DP
25°	.405/DP

Fig. 2

circular pitch, a standard proportioned gear set is developed. If the pinion tooth thickness used is greater than half of the circular pitch, the pinion will be oversized.

There are many causes for gear tooth failure, as shown in AGMA 110.04 "Nomenclature of Gear Tooth Wear and Failure." The two principal modes for gear tooth failure are tooth breakage by bending fatigue and surface failure by pitting. More advanced failure considerations include scoring, spalling, rolling, peening, rippling, case crushing and various forms of wear.

The program segment for calculating the horsepower ratings uses the procedure outlined in AGMA 218.01 "For Rating the Pitting Resistance and Bending Strength of Spur and Helical Involute Gear Teeth" and is based on the stresses caused by the contact or compressive stresses where the gear teeth meet, and on the bending or tensile stresses which occur in the gear fillet area. See Fig. 3. Examples of these two failures are shown in Fig. 4.

The use of the superimposed parabola to determine the critical fracture or fatigue point on the root fillet is shown in Fig. 5 and is used to compute the interim values needed to establish the bending power rating.

The surface stresses caused by the rolling action of contacting cylinders results in elastic deformations. If the subsurface shear that develops exceeds the strength of the material, a crack occurs and propagates up to the surface developing a pit. See Fig. 6.

Fig. 7 illustrates the pattern of tooth pair contact for a gear with a contact ratio of about 1.5. In the central part of the tooth form a single pair of teeth are in contact. For shared load the highest point of single tooth contact is used in calculating the beam strength and the lowest point of single tooth contact is used in the surface strength calculation. The former is used for the "J" factor, and the latter is used for the "I" factor.



Fig. 3

Fig. 4 – Types of stress failures on a gear tooth

a – Destructive pitting

b – Spalling

The AGMA equations for the bending and pitting power ratings are shown below, first with all the K and C factors and, then, in simplified form without these factors.

$$P_{AT} = \frac{np\ d\ Kv}{126{,}000\ K_A} \bullet \frac{F}{Pd} \bullet \frac{J}{K_S K_M} \bullet \frac{S_{AT}\ K_L}{K_R\ K_T}$$

np = Pinion rpm
d = Pinion oper. p.d.
F = Gear face
Pd = Diametral pitch
J = Geom. factor-bending
S_{AT} = Tensile stress no.

$$P_{AT} = \frac{np\ d}{126{,}000} \bullet \frac{F}{Pd} \bullet J \bullet S_{AT} \qquad (1)$$

$$P_{AC} = \frac{np\ F}{126{,}000} \bullet \frac{I\ Cv}{C_S\ C_M\ C_F\ C_A} \bullet \left(\frac{d\ S_{AC}}{Cp} \bullet \frac{C_L\ C_H}{C_T\ C_R}\right)^2$$

np = Pinion rpm
F = Gear face
I = Geom. factor – pitting
d = Pinion Oper. p.d.
S_{AC} = Surface comp. stress no.
Cp = Elastic coeff.

$$P_{AC} = \frac{np \bullet F \bullet I}{126{,}000} \bullet \left(\frac{d \bullet S_{AC}}{Cp}\right)^2 \qquad (2)$$

The factors can generally be used as 1.0 and can mainly be considered as warning flags to induce some thought on the part of the designer. The flags are as follows:

- Ka & Ca – application factor which considers the even or shocky nature of the prime mover and the absorbing load;
- Kv & Cv – related to the effect that dynamically induced loads might cause, usually due to higher velocities;



Fig. 5



Fig. 6



Fig. 7

- Ks & Cs – used in consideration of the effect that the actual physical size of the teeth might have;
- Km & Cm – warn about the effect of improper load distribution across the gear face;
- Kl & Cl – related to gear life. Normal life is planned for many millions of stress cycles and is modified for shorter life needs. See Fig. 8
- Kr & Cr – reliability factors normally expecting less than 1 failure in 100;
- Kt & Ct – temperature factors warning that gears normally do not exceed 250°F;
- Cf – consideration factor for the effect of surface finish on the surface strength;
- Ch – is a factor for the ratio of hardness between the gear and pinion.

Again, for simplicity the K & C factors will be used as 1.0, although some consideration has to be given to the application factor Ka & Ca and the reliability factor Kr & Cr in the final selection, as they are important and significant.

For the gear material chosen, steel at 300 BHN, the AGMA suggests the allowable tensile stress number as 40,000 and the allowable compressive stress number as 130,000 for input.

The complete computer output for one gear set design is shown in Table 5. At this point we can tell if the power capacity of these designs will meet the needs and with what reliability or safety factor.

Noise Considerations

Gear noise is complex and difficult to analyze, but is usually related to gear accuracy. For equivalent accuracy among gears with various pressure angles, it has been observed that lower pressure angle gears are quieter and, conversely, higher pressure angles are noisier. The relationship is shown in Fig. 9. If we have sufficient capacity in the gears designed, we could choose the lower pressure angle for a quieter set. This does not mean sets with higher pressure angles cannot be quiet, but they will probably require a higher level of gear accuracy.



Fig. 8

Table 5 – Complete Computer Output for One Gear Set Design

Center Distance	*5.0000*	
Gear Ratio	4.0000	
Face Width	0.800	
Diametral Pitch	10.0000	
Pressure Angle	20.0000	
Pinion RPM	1800.00	
Tensile Stress No.	40000	
Compressive Stress No.	130000	
Backlash	0.0020	
Contact Ratio	1.691	
	Pinion	*Gear*
Teeth	20.0	80.0
Outside Diam.	2.200	8.200
Pitch Diam.	2.0000	8.0000
Oper. P.D.	2.0000	8.0000
Root Diam.	1.750	7.744
Base Diam.	1.8794	7.5175
Cir. Tooth Thick.	0.1571	0.1551
Tip Flat	0.0695	0.0778
T.I.F. Diam.	1.8850	7.8547
J FACTOR	0.368	
Horsepower Beam	*33.6*	
Horsepower Surf.	15.9	

FAST.

The CNC controlled HNC-35 Worm and Thread Grinder

Klingelnberg introduces the fastest . . . most accurate complex worm and thread grinder available . . . anywhere in the world! The CNC controlled HNC-35 stores data for up to 960 complex shapes, reducing set-up to 10-15 minutes . . . a fraction of the time required with mechanically controlled grinders.

The HNC-35 is versatile too. It performs creep-feed grinding from the solid, eliminating the need for preliminary milling of worms, threads and rotors. It's two machines in one . . . with the high degree of accuracy you'd expect from Klingelnberg.

The HNC-35 is available with a mechanical dresser or an optional CNC dresser, for special forms and flanks. Whether you produce small quantities or long production runs, the FAST set-up . . . FAST cycling . . . HNC-35 will improve your worm and thread productivity.

For additional information and a copy of our catalog, contact: Klingelnberg Corporation, 15200 Foltz Industrial Parkway, Cleveland, OH 44136. Or, phone (216) 572-2100 for an extra FAST response.

KLINGELNBERG



CIRCLE A-9 ON READER REPLY CARD

Another area of consideration if noise reduction is of great importance is the use of non-standard working depth gears, such as stub or shorter working depth or extended working depth. See Fig. 10.

Stub teeth are very rigid and tend to be noisier than the longer extended tooth forms where more tooth flexure is present. *These tooth depth systems require separate geometrical* considerations for design especially in regard to the minimum number of teeth.

Strength Considerations

If maximum strength is of significant importance, then higher pressure angles would be chosen, as is shown in Fig. 11. Here the relationship is shown for strength versus pressure angle, where the strength increases with increasing pressure angle. This is also confirmed later using the AGMA horsepower capacity calculations.

For the gear material chosen, the computations for horsepower show the bending strength of the pinion is greater than the surface strength. This is quite typical of steel or other materials with insufficient surface strength. Typically such gears wear out rather than failing by fracture. It is preferable, obviously, to have a gear set fail progressively by wear rather than catastrophically by breakage.

The results of some 15 different program runs are summarized in Table 6. The horsepower based on bending and surface strength is tabulated. Several things can be noted. First, in the pitch range chosen all the sets have higher bending strength than surface strength. Second, the surface strength goes up as the teeth get smaller, and third, the bending strength goes down as the teeth get smaller. Fig. 12 presents a graphical comparison of the computations for the 20°PA group. In essence the surface strength is fairly level across the entire pitch range while the bending strength decreases with the finer pitches.

Reliability

To reduce the chance of failure in the gear set, the Kr & Gr factor as suggested by AGMA can be considered. The factor for high reliability, less than 1 in 1000 failures, is 1.25,



WORKING DEPTH

Fig. 10



COMPARISON OF STRENGTH WITH PRESSURE ANGLE

Fig. 11



COMPARISON OF NOISE WITH PRESSURE ANGLE

Fig. 9



Fig. 12

(continued on page 42)

GEAR DESIGN OPTIONS . . .
(continued from page 18)

Table 6

PA	DP	HPC	HPB
14.5°	10	12.7	26.7
	12	13.4	24.1
	14	13.8	22.0
	16	14.0	20.2
	20	14.3	17.3
20°	10	15.9	33.6
	12	16.7	30.0
	14	17.1	27.1
	16	17.5	24.6
	20	17.9	20.8
25°	10	18.8	40.3
	12	19.6	35.8
	14	20.1	32.1
	16	20.5	29.0
	20	21.0	24.4

and that for very high reliability or less than 1 in 10,000 failures is 1.5. The computed capacity would be reduced in accordance with the degree of reliability desired.

Comments

The opportunity to use a computer to assist in gear design can present the designer with a large choice of possible candidates. Exercising options and choices can help in zeroing in on a final selection, but obviously there is no one design that fits all requirements. Many different gear sets can be suitable.

Designers are not restricted to standard pressure angles or pitches nor to standard tooth forms. Appropriate basic data can be placed in the program if desired, for example, to use existing tooling even if metric module, to reduce costs or time delays.

No gear design is really considered final or so perfect that the parts can go into production without some model testing or pilot manufacturing. It is at this point that the unforseen factors can be dealt with.

References:

1. AGMA 110.04-1980, "Nomenclature of Gear Tooth Failure Modes." American Gear Manufacturers Association, Arlington, Virginia.
2. AGMA 218.01-1982, "For Rating the Pitting Resistance and Bending Strength of Spur and Helical Involute Gear Teeth." American Gear Manufacturers Association, Arlington, Virginia.

This article was first presented at the SME Gear Processing and Manufacturing Clinic, Schaumburg, IL Nov. 11-13, 1986.

Gear Design Program

```
10 REM spurgear.bas
20 CLS:KEY OFF
30 PI=3.141592654#:RA=180/PI
40 INPUT "Diametral Pitch          =",DP"
50 INPUT "Pressure Angle"          =",PA
60 INPUT "Center Distance"         =",CD
70 INPUT "Face Width               =",F
80 INPUT "Steel Allow Tens.        =",SAT
90 INPUT "Steel Allow Comp.        =",SAC
100 INPUT "Pinion RPM              =",RPM
110 INPUT "Pinion Teeth            =",NP
120 INPUT "Gear Teeth              =",NG
130 INPUT "Pinion CTT              =",CTP
140 INPUT "Backlash                =",BL
150 PRINT"     Program in progress, turn printer on, please."
160 HADD=1.25/DP:TIPR=.3/DP:HDED=1/DP:CLR=.25/DP
170 HTF=1.5708/DP-2*HADD*TAN(PA/RA)
180 TCD=(NP+NG)/2/DP
190 PDP=NP/DP:PDG=NG/DP:PRP=PDP/2:PRG=PDG/2
200 BDP=PDP*COS(PA/RA):BDG=PDG*COS(PA/RA)
210 OPPDP=2*CD*NP/ (NP+NG):OPPDG=2*CD*NG/(NP+NG)
220 OPPRG=OPPDG/2:OPPRP=OPPDP/2
230 CP=PI/DP:CSP=CP-CTP:BRG=BDG/2:BRP=BDP/2:BP=CP*COS(PA/RA)
240 PDED=(CSP-HTF)/2/TAN(PA/RA)
250 RDP=PDP-2*PDED
```

(continued on page 44)

Three Week Hobs

Barber-Colman Announces PDQ Hobs

Three week delivery of special-form hobs!

We're stocking machined blanks, covering the DP range of 3 to 100, ready for finish grinding. (Refer to chart below.)

Call our toll-free hot line (800/435-4176) with *your order and tooth form details. Or send* us your tooth form by FAX (815/877-0264). We'll CAD design the cutting tool, form the grinding wheel, and ship the tools in 2-1/2 weeks.

Scope: The program covers gear hobs, involute spline hobs, straight sided serration hobs, and parallel key spline hobs. Not covered are shanks, tapered bores or periphery, or clutch keyways. And, there may be times we'll have to limit your order size to assure fast delivery for everyone... but we'll always be able to start you cutting parts in three weeks!



PDQ BLANKS

The Barber-Colman quality you've come to expect — Now it's yours in THREE WEEKS

TYPICAL DP RANGE	DIAMETER	LENGTH	BORE	MATERIAL
35–100	.750	.500	.315	M42
30–80	1.125	1.125	.500	M42
16–35	1.875	1.875	.750	M42
8–20	3.000	3.000	1.250	M3
4–10	4.000	4.000	1.250	M3
3–6	5.000	5.000	1.500	M3

Now Available our full color brochure.

For further information call us today!

BARBER COLMAN COMPANY

Barber-Colman Company
SPECIALTY TOOLS DIVISION

F 22888

LITHO IN U.S.A.

CIRCLE A-17
ON READER REPLY CARD

1351 Windsor Road
P.O. Box 2950
Loves Park, IL U.S.A. 61132-2950

Phone: (815) 877-0241
TLX: 4937737 BCSTUI
FAX: 815-877-0264

GEAR DESIGN PROGRAM

```
260 ODG=2*CD-RDP-2*CLR:ORG=ODG/2
270 ODP=RDP+2*(HADD+HDED):ORP=ODP/2
280 COSOP=TCD*COS(PA/RA) /CD
290 OPPA=RA*ATN(SQR(1-COSOP^2)/COSOP)
300 INVDP=TAN(OPPA/RA)-(OPPA/RA)
310 INVPA=TAN(PA/RA)-(PA/RA)
320 CTG=CP+2*TCD*INVDP-2*TCD*INVPA-CTP-BL
330 GDED=(CP-CTG-HTF)/2/TAN(PA/RA)
340 RDG=PDG-2*GDED
350 WDG=(ODG-RDG)/2:WDP=(ODP-RDP)/2
360 PX=BDP/ODP:XA=ATN(SQR(1-PX^2)/PX)
370 INVXA=TAN(XA)-XA
380 GX=BDG/ODG:XB=ATN(SQR(1-GX^2)/GX)
390 INVXB=TAN(XB)-XB
400 TFP=ODP*(CTP/PDP-INVXA+INVPA)
410 TGF=ODG*(CTG/PDG-INVXB+INVPA)
420 LA=SQR(CD^2-(BRG+BRP)^2)
430 LAP=LA-SQR(ORG^2-BRG^2)
440 LAG=LA-SQR(ORP^2-BRP^2)
450 WLRP=SQR((LAP+BP)^2+BRP^2)
460 TIFRP=SQR(BRP^2+LAP^2)
470 TIFRG=SQR(BRG^2+LAG^2)
480 CR=(LA-LAP-LAG)/BP
490 Y=PDED-TIPR
500 X=CTP/2+Y*TAN(PA/RA)+TIPR/COS(PA/RA)
510 Q=BRP/WLRP:WLPA=ATN(SQR(1-Q^2)/Q):INVWL=TAN(WLPA)-WLPA
520 B=CTP/PDP-INVWL+INVPA:A=WLPA-B
530 DL=BDP/COS(A)
540 PRINT"     Program in progress."
550 AL=PI/4
560 E=(X+Y/TAN(AL))/PRP
570 BA=AL-E
580 KS=Y/SIN(AL):KE=KS+TIPR
590 TE=PRP*SIN(E)-(KE*COS(BA))
600 NE=PRP*COS(E)-(KE*SIN(BA))
610 HH=DL/2-NE
620 YY=2*HH*TAN(BA)-TE
630 IF ABS(YY)<.0001 Then 670
640 YYY=(2*HH/(COS(BA)^2)-KE*SIN(BA))*(1-KS/PRP/SIN(AL))+KS*SIN(BA)
650 AL=AL-YY/YYY
660 GOTO 560
670 H=HH:TH=TE*2
680 X1=TH^2/4/H
690 Y2=DP/((COS(A)/COS(PA/RA))*((1.5/X1)-TAN(A)/TH))
700 RM=Y^2/(Y+PRP)+TIPR
710 H5=.18-.008*(PA-20)
720 J5=H5-.03
730 L5=.45+.01*(PA-20)
740 K5=H5+(TH/RM)^J5*(TH/H)^L5
750 J=Y2/K5
760 HPB=RPM*OPPDP*F*J*SAT/126000!/DP
770 R1=LA-LAG-BP:R2=LA-R1:R3=SIN(OPPA/RA)*OPPRP:R4=LA-R3
780 CX=R1*R2/R3/R4
790 CG=NG/(NG+NP)
```

GEAR DESIGN PROGRAM

```
800 CC=COS(OPPA/RA)*SIN(OPPA/RA)*CG/2
810 I=CC*CX
820 HPS=RPM*F*I*(OPPDP*SAC/2300)^2/126000!
830 IF LAP<0 THEN CR=0:TIFRP=0
840 LPRINT"         Gear Design Summary"
850 LPRINT
860 LPRINT
870 A$="Center Distance              ###.####"
880 LPRINT USING A$;CD
890 B$="Gear Ratio              ##.####"
900 LPRINT USING B$;NG/NP
910 C$="Face Width              ##.###"
920 LPRINT USING C$;F
930 D$="Diametral Pitch         ###.####"
940 LPRINT USING D$;DP
950 E$="Pressure Angle          ##.####"
960 LPRINT N USING E$;PA
970 F$="Pinion RPM              #####.##"
980 LPRINT USING F$;RPM
990 G$="Tensile Stress No.      ######."
1000 LPRINT USING G$;SAT
1010 H$="Compressive Stress No.  ######."
1020 LPRINT USING H$;SAC
1030 I$="Backlash                #.####"
1040 LPRINT USING I$;BL
1050 J$="Contact Ratio           #.###"
1060 LPRINT USING J$;CR
1070 LPRINT
1080 LPRINT"            Pinion          Gear"
1090 K$="Teeth           ###.#           ###.#"
1100 LPRINT USING K$;NP;NG
1110 L$="Outside Diam.   ###.###         ###.###"
1120 LPRINT USING L$;ODP;ODG
1130 M$="Pitch Diam.     ###.####        ###.####"
1140 LPRINT USING M$;PDP;PDG
1150 N$="Oper. P.D.      ###.####        ###.####"
1160 LPRINT USING N$;OPPDP;OPPDG
1170 0$="Root Diam.      ###.###         ###.###"
1180 LPRINT USING 0$;RDP;RDG
1190 P$="Base Diam.      ###.####        ###.####"
1200 LPRINT USING P$;BDP;BDG
1210 Q$="Cir. Tooth Thick.  #.####          #.####"
1220 LPRINT USING Q$;CTP;CTG
1230 R$="Tip Flat           #.####          #.####"
1240 LPRINT USING R$;TFP;TFG
1250 S$="T.I.F. Diam     ###.####        ###.####"
1260 LPRINT USING S$;TIFRP*2;TIFRG*2
1270 T$="J FACTOR         #.### "
1280 LPRINT USING T$;J
1290 U$="Horsepower Beam  ####.#         ####.#"
1300 LPRINT USING U$;HPB
1310 V$="Horsepower Surf.  ####.#        ####.#"
1320 LPRINT USING V$;HPS
1330 IF CR=O THEN LPRINT" Warning! Possible undercut on pinion. Suggest design change."
```